/*
Copyright (c) 2011, Felipe Michels Fontoura
All rights reserved.

Redistribution and use in source and binary forms, with or without
modification, are permitted provided that the following conditions are met:
    * Redistributions of source code must retain the above copyright
      notice, this list of conditions and the following disclaimer.
    * Redistributions in binary form must reproduce the above copyright
      notice, this list of conditions and the following disclaimer in the
      documentation and/or other materials provided with the distribution.
    * Neither the name of Felipe Michels Fontoura nor the names of his
      contributors may be used to endorse or promote products derived from
      this software without specific prior written permission.

THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS" AND
ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE IMPLIED
WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE ARE
DISCLAIMED. IN NO EVENT SHALL FELIPE MICHELS FONTOURA BE LIABLE FOR ANY
DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR CONSEQUENTIAL DAMAGES
(INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR SERVICES;
LOSS OF USE, DATA, OR PROFITS; OR BUSINESS INTERRUPTION) HOWEVER CAUSED AND
ON ANY THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR TORT
(INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE OF THIS
SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.
*/

/* Crenças */

/* verifica se o robô está em uma organização */
emOrganizacao :- orgPoeira(_, _, [_, _, _]).
emOrganizacao :- orgOleo(_, _, [_, _]).

/* verifica se deve varrer */
devo(varrerPoeira) :- sou(Eu) & orgPoeira(_, _, [Eu, _, _]).
devo(segurarPa)    :- sou(Eu) & orgPoeira(_, _, [_, Eu, _]).
devo(carregarLixo) :- sou(Eu) & orgPoeira(_, _, [_, _, Eu]).

/* unifica a posição de alvo */
alvo(X, Y) :- orgPoeira(X, Y, [_, _, _]).
alvo(X, Y) :- orgOleo(X, Y, [_, _]).

/* verifica se todos estão juntos */
todosJuntos :- orgPoeira(X, Y, [R1, R2, R3])
             & estaEm(R1, X, Y) & estaEm(R2, X, Y) & estaEm(R3, X, Y).
todosJuntos :- orgOleo(X, Y, [R1, R2])
             & estaEm(R1, X, Y) & estaEm(R2, X, Y).

/* Objetivos iniciais */

!iniciar.

/* Planos */

/* iniciar o funcionamento, dizendo para o orgManager que está livre */
+!iniciar : true
	<- .my_name(Eu); +sou(Eu);
	   .send(orgManager, tell, estou(Eu, livre));
	   !descreverConhecimentos.

/* informar para o orgManager tudo o que sabe fazer */
+!descreverConhecimentos : sei(Fazer) & not gerenteSabeQueSei(Fazer)
	<- ?sou(Eu);
	   .send(orgManager, tell, sei(Eu, Fazer));
	   +gerenteSabeQueSei(Fazer);
	   !descreverConhecimentos.
+!descreverConhecimentos : true
	<- !!decidir.

/* decidir o que fazer. */
+!decidir : not emOrganizacao
	<- nop; !!decidir.
+!decidir : alvo(X, Y) & not estouEm(X, Y)
	<- !ir(X, Y); !!decidir.
+!decidir : alvo(X, Y) & estouEm(X, Y) & not todosJuntos
	<- nop; !!decidir.
+!decidir : alvo(X, Y) & estouEm(X, Y) & todosJuntos
	<- !executarFuncao.

/* executar a função na organização */
+!executarFuncao : devo(ExecutarAcao) & not sei(ExecutarAcao)
	<- .print("ERRO!"); !liberar.
+!executarFuncao : devo(ExecutarAcao)
	<- ExecutarAcao; !liberar.
-!executarFuncao : true
	<- nop; !!executarFuncao.

/* liberar o robô da organização */
+!liberar : orgPoeira(X, Y, [R1, R2, R3])[source(Fonte)]
	<- -orgPoeira(X, Y, [R1, R2, R3])[source(Fonte)];
	   ?sou(Eu); .send(orgManager, tell, estou(Eu, livre));
	   !!decidir.
+!liberar : orgOleo(X, Y, [R1, R2])[source(Fonte)]
	<- -orgOleo(X, Y, [R1, R2])[source(Fonte)];
	   ?sou(Eu); .send(orgManager, tell, estou(Eu, livre));
	   !!decidir.

/* ir para uma posicao */
+!ir(X, Y) : estouEm(X, Y)            <- true.
+!ir(X, Y) : estouEm(Xr, Y)  & X > Xr <- !darPasso(direita);  !ir(X, Y).
+!ir(X, Y) : estouEm(Xr, Y)  & X < Xr <- !darPasso(esquerda); !ir(X, Y).
+!ir(X, Y) : estouEm(Xr, Yr) & Y > Yr <- !darPasso(baixo);    !ir(X, Y).
+!ir(X, Y) : estouEm(Xr, Yr) & Y < Yr <- !darPasso(cima);     !ir(X, Y).

/* dar um passo em uma direcao */
+!darPasso(D) : not olhandoPara(D) <- olharPara(D); !darPasso(D).
+!darPasso(D) : true               <- darPasso.